
Introduction


Internet applications are client/server applications that require relatively lightweight servers. Otherwise, the servers will be swamped. Intranet applications, by contrast, are intended for a more limited user base and might very well afford more computationally intensive applications. One way of implementing such applications is to use JAVA on the client side and a more traditional industrial strength programming language on the server side. The use of JAVA applets gives the application a high degree of machine independence and permits it to within Netscape or Internet Explorer. It is also more powerful than clients running purely from HTML hot links. The server side, provided that it can communicate effectively with the client, could be implemented in Smalltalk.





There are two ways of doing this. One approach requires the use of a Smalltalk environment that can generate JAVA bytecodes so that Smalltalk can be used on both sides. Another way requires the use of a JAVA library specially designed to support Smalltalk applications running on a server. Such a client/server package is being developed by Applied Reasoning Systems for ParkPlace/Digitalk’s VisualWorks Smalltalk under the name “Classic Blend”. In this column, we will develop an internet application using a beta version of their software.


A Server-Side Browser for Downloading Files


Currently, files provided by internet sites for downloading are typically listed as hot-linked text items in HTML documents. Clicking on such a file item generally invokes the client’s file browser so that the user can specify where to save the file. We would like to provide a similar facility on the server side; i.e., we want to permit users to browse the server’s public directories using a generic file browser. 





Given an arbitrary web site with a hot-link reference such as “Browse Files” (see Figure 1), we would like to get a file browser (see Figure 2) that we can use to view the public directories. By clicking on “Download”, simple text and graphics files might be displayed below the browser. Such files can be explicitly saved by users using Netscape (Explorer) facilities. By contrast, clicking “Download” for non-displayable files such as zip files would immediately cause the host’s file browser to appear (see Figure 3) so that the destination directory might be specified.


Reusing Existing Facilities


A little over a year ago, we designed and implemented an enhanced browser to replace the aging VisualWorks’s file browser [1]. It was designed to be executed simply via an expression such as “FileDialog new open” to bring up a dialog as shown in Figure 4. The user navigates the file system by (1) double-clicking on a directory to either expand it or contract it, depending on its current state, or by (2) typing in the name of a search path (with or without current directory relative notation such as “..\directory1\directory2\*.txt” or “world.*”). Once the desired file is located, it can be retrieved by double-clicking on the file name or by clicking on the OK button. The full path name of the file (as a string) is returned. Clicking on Cancel causes nil to be returned instead. Built-in filters (obtainable through the drop-down menus at the bottom-left) are provided to narrow the focus of a user’s search. User specified filters can also be provided at open time, for example, as follows:





		FileDialog new


				openOn: '*' 


				title: 'Open' 


				fileTypes: #('All files (*)' 'Smalltalk files (*.st)')





It is also possible to target specific disk drives using the associated drop-down menu at the bottom right. 





Now, given this dialog facility, our goal is to rework it as an internet application. But there is one small constraint. The beta version of  “Classic Blend” doesn’t support drop-down lists. But that’s easily fixed using the window builder to replace the two drop-down lists by standard list views. Rather then simply modify the FileDialog class, we decided to subclass it as HTMLFileDialog. Our modified windowSpec could then be saved in the new class rather than modifying the existing one.


Running a “Test” Internet Application


But before we embarked on this route, we thought it would be safer to use our existing “untouched” file dialog as our first test of an internet application. We placed the “Classic Blend” JAVA libraries in Netscape subdirectories and created a simple HTML document to launch our application. This file (see Listing 1) contains the name “HTMLFileDialog” as the applet type and “windowSpec” as the spec name. To start the server’s object request broker, we executed





		OrbServerModel open





in a special version of Smalltalk that contained the “Classic Blend” extensions. Then we dragged the HTML file of Listing 1 into Netscape to test it. The alternative was to provide a hot-link to this file in one of our standard web pages.





To our surprise and delight, a Netscape “version” of our file dialog came up. Not unexpectedly, the pull-down lists were missing. The entry field for targeting searches worked as expected. However, we couldn’t open/close directories. We concluded that VisualWorks was interpreting double-clicks in Netscape as two single click events because communication over the net was slow. VisualWorks doesn’t explicitly receive double-click events from the operating system it is running on. It simply interprets successive single-click events as a double-click event if the time between the clicks is sufficiently short. When we decided we were done, we clicked on the OK button but nothing happened. The Cancel button didn’t work either. After showing this to a few fellow programmers, they laughed at our naiveté. Did we really expect Netscape to go back to the previous page, they chucked! The “Classic Blend” facilities very nicely disabled our buttons (we didn’t have to do anything to get that effect).


Designing the Server-Side File Browser


We didn’t want to make any changes to the existing FileDialog class but we couldn’t resist fixing a number of small problems that were irrelevant to the our extension. Since the majority of the code was provided in the Sept ’95 JOOP column [1], we won’t repeat it here. 





We started by using the window builder to redesign the layout of the existing file dialog. We replaced the OK and Cancel buttons by a Download button, replaced the file filter drop-down list by a standard list box, changed the name of the holder for the read-only text box that displays the currently selected directory (we’ll see why later), removed the double-click notification messages from the directory list box, and deleted the drop-down list for disk drives. This last point is one of the more important changes that we had to make. Because our existing file dialog permits a user to target any file on the server, we wanted to “hardwire” users to a specific public directory.





We started by providing a class (and instance) method webRoot (see Listing 2) to specify the public web-accessible directory. The actual name of this root directory was to be invisible to users. But the original file dialog always displayed the full path name of the selected directory. By using a new holder webDirectoryHolder for this read-only text box, the contents of the original holder currentDirectoryHolder which is still being used by the application, is not visible. Our update method transfers the contents of currentDirectoryHolder to webDirectoryHolder but replaces the web root by of 'web::'. 





Because the original file dialog lists all directories contained by a targeted disk drive in an indented fashion (contained directories are indented under the parent), the number of spaces preceding each name depends on the depth of the hierarchy. In the subclass, some of the parents are omitted because they are part of the web root. Without changing the implementation, we get an excessive number of “apparently” unnecessary spaces if the web root is long. So we added a method indentedOffset in the superclass which simply returns zero. The subclass returns webIndentedOffset which is the negative of the number of directories (counting the drive as one) in the web root.


  


The existing file dialog has a built-in history facility that “remembers” the context of the last browsing session for each disk drive. A number of “do nothing” methods (defaultDisk, loadForDisk:, and storeForDisk:) were specifically added to disable this history facility. To make that work, we had to override method initialUpdate which sets up the initial starting files (files below the web root but not the root itself) using facilities provided by the superclass. Navigation is done by expanding/contracting existing files. Full path names are manipulated in the superclass but only the last name of each file is displayed. We also provided an initialize method that adds a number of useful “networking” filters.





The most complex methods are reaction methods that handle user requests; e.g., typedNewFileName, changedDirectory, and changedFileType. These make heavy use of inherited methods which you’ll have to accept on faith. But note that typedNewFileName only permits a file name that matches the web root. The most interesting method, however, is download which does nothing if the web server (the orb) isn’t running. It makes use of a “Classic Blend” extension surfTo:inFrameNamed: which can send a file name to an HTML “frame”. If the file is a text or graphics file (a “.gif”) file, it is simply displayed in this frame. The Netscape user can then copy this file to a local target directory with standard Netscape facilities. If the file is not displayable, Netscape provides a host file browser to permit the user to choose the target directory.


Conclusions


We used Applied Reasoning Systems’ “Classic Blend” facilities to implement a server-side browser in VisualWorks Smalltalk. The application’s window builder specification is used to transparently create an equivalent application that runs on the client-side as a JAVA applet. Communication between the applet and the Smalltalk application is also transparently managed through a TCP/IP connection. The use of JAVA applets and a direct connection provide better security than most other approaches and also permits better performance and better application layout control than HTML based applications.

Where to Obtain the Code


Source code for this article can be obtained on the World Wide Web at http://www.objectpeople.com. It is implemented in ParkPlace/Digitalk’s VisualWorks and uses Applied Reasoning Systems’s “Classic Blend”.

 Figure 1  A hot-link to a server-side file browser.

 Figure 2  Viewing public directories on a server.

 Figure 3  Using the host’s file browser to specify a download target directory.

 Figure 4  Using an instance of FileDialog.

 Figure 5  After starting the web server.





Listing 1 File filebrowser.html.





<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 3.2//EN">


<HTML>


<HEAD>


   <TITLE></TITLE>


   <META NAME="Author" CONTENT="Wilf LaLonde">


</HEAD>


<BODY>





<P><B>File Browser</B></P>





<TABLE BORDER=3 >


<TR>


<TD>


<APPLET code=ars.cb.ifc.support.IFCApplet width=400 height=350>


	<PARAM name="ApplicationClass" value="ars.cb.ifc.support.IFCApplication">


	<PARAM name="host" value="localhost">


	<PARAM name="port" value="12345">


	<PARAM name="appletType" value="HTMLFileDialog">


	<PARAM name="specName" value="windowSpec">


	<PARAM name="bindings" value="file:../classBindings.txt">


	<PARAM name="appletName" value="fileBrowser">


	<PARAM name="orbName" value="fileBrowser">


	<PARAM name="debug" value="false">


</APPLET>


</TD>


</TR>


</TABLE>





</BODY>


</HTML>





Listing 2 The HTMLFileDialog class.





class:									HTMLFileDialog


superclass:							FileDialog


instance variables:			webDirectoryHolder


											webDirectoriesHolder 


											webFileNamesHolder 


											webFileTypesHolder”





class methods 





examples


example1


	"HTMLFileDialog example1"


	^HTMLFileDialog new open





example2


	"HTMLFileDialog example2"


	^HTMLFileDialog new


		openOn: '*' 


		title: 'Open' 


		fileTypes: #('All files (*)' 'Smalltalk files (*.st)')





web setup


webRoot


	^'c:\Applications


webRootPattern


	^'c:\Applications\*'


webDisk


	^'c:'


webIndentedOffset


	^-2 "The negative of the number of $/ characters in the web root pattern."





querying


defaultDisk


	"This method disables the superclass default."


	^nil





interface specs (generated by window builder)


windowSpec


	"… code not shown ..."





instance methods





instance initialization


initialize


	super initialize.


	fileTypes := #(


		'All files (*)' 


		'Word files (*.doc)'


		'Text files (*.txt)'


		'Zip files (*.zip)'


		'Gif files (*.gif)'


		'Smalltalk files (*.st)' 


		'Change log files (*.cha)').


loadForDisk: desiredDisk


	"This method disables the history feature."


storeForDisk: diskName


	"This method disables the history feature."





web facilities


webRoot


	^self class webRoot


webRootPattern


	^self class webRootPattern


webDisk


	^self class webDisk


indentedOffset


	^self class webIndentedOffset





matchesWebRoot: path


	| root |


	root := self webRoot.


	^path size >= root size and: [


		(path copyFrom: 1 to: root size) sameAs: root]





rootlessPath: path


	| suffix |


	^(self matchesWebRoot: path)


		ifTrue: [


			suffix := path copyFrom: self webRoot size + 1 


				to: path size.


			suffix isEmpty 


				ifTrue: [''] 


				ifFalse: [suffix copyFrom: 2 to: suffix size]]


		ifFalse: ['']





URLForFile: fileName


	| drive path |


	drive := fileName copyFrom: 1 to: 1.


	path := fileName copyFrom: 3 to: fileName size.


	^'file:///', drive, '|', path





expanding/contracting


expandDirectory: directory


	| paths |


	paths := self directoriesMatching: directory, 


		self separatorString, '*'.


	self addDirectoryPaths: paths 


		before: directories size + 1





reacting


typedNewFileName


	"React to hitting enter (or losing focus) after typing a new file name."


	| fullPattern nonDirectory |


	pattern := self sanitizeSeparators: 


		self filePatternHolder value.


	fullPattern := self fullPathName: pattern.


	(self matchesWebRoot: fullPattern) ifFalse: [^self].


	self selectPattern: fullPattern.


	self ensureExpanded.


	nonDirectory := self nonDirectoryPortionOf: fullPattern.


	nonDirectoriesIndex := self indexOf: nonDirectory 


		in: self fileNamesHolder list.


	self update





changedDirectory


	"React to single-clicking an entry in the directory list. Don't permit deselecting. Expand new selections; contract old ones."


	| newIndex wasExpanded |


	newIndex := self directoriesHolder selectionIndex.


	(newIndex > 0) & (directoriesIndex ~= newIndex)


		ifTrue: [


			directoriesIndex := newIndex.


			self reselectDirectory.


			self ensureExpanded]


		ifFalse: [


			wasExpanded := 


				self isExpandedAtIndex: directoriesIndex.


			self reselectDirectory. 


			self ensureFlippedFrom: wasExpanded].


	self update





changedFileType


	self fileTypesSelectionHolder 


		value: self webFileTypesHolder selection.


	super changedFileType





closeCancel


	"React to clicking on the CLOSE box."


	self close





download


	"Download if everything is set up to work."


	OrbServer currentServer isConnected ifFalse: [^self].


	result := self okResult.


	(self containsWildCard: result)  ifTrue: [


		^Dialog warn: (self rootlessPath: result), 


			' can''t be downloaded'].


	self builder window 


		surfTo: (self URLForFile: result)


		inFrameNamed: 'main'





updating


initialUpdate


	super initialUpdate.


	self expandDirectory: self webRoot.


	self currentDirectoryHolder value: self webRoot.


	self webFileTypesHolder 


		list: fileTypesHolder value; 


		selection: fileTypesSelectionHolder value





update


	super update.


	self webDirectoryHolder value: 'web::', 


		(self rootlessPath: self currentDirectoryHolder value)





aspects (generated by window builder)


webDirectoryHolder


	"… code not shown ..."


webFileNamesHolder


	"… code not shown ..."


webFileTypesHolder


	"… code not shown ..."

